T102

USER MANUAL

TABLE OF CONTENTS

I. Features:

Time of day in Hours Minutes and Seconds.
24 hour time format.
Month and Day date function.
On-board crystal controlled time base.
Simple to use latched BCD interface.
Easy interface to BASIC or other languages.
Fast set, slow set, and register reset functions allow rapid setting of clock and date.
Optional battery backup capability.
Backup voltage may be 8 to 18 volts, unregulated DC.
Low battery drain (5-15ma) in backup mode.
Can be located at any group of 4 I/O port address.

II. General:

Before installing the CT102-A board in your system the address must be selected by setting the switches on SW1. After deciding which I/O port address will be used to access the board in your system, look up the address you have selected in table II-2 and set the switches as shown. Be sure that switches 7 and 8 are open while initially testing the board. Do not install any options until you have tested the board. Inspect the board for shorts or damaged components. Insure that all the IC's are properly installed with pin 1 toward the S100 connector. Also check that the proper IC type is in each location.

Carefully install the board in the system with power off, being certain that pins 1 and 50 line up with the ground pins on the s100 bus connector. Plugging the board in backwards will cause damage to the board. After powering on the system you may test the CT102-A as follows:

1. Enter the I/O address you have selected for the board and write a HEX 22(decimal 34) to the board. The fast set mode will be selected and the minutes tens digit of the clock will be selected for input. If necessary you may enter an I/O instruction in memory and execute it.
2. Read the I/O address and the value read will be seen to be changing at better than a 1Hz rate. If you have a front panel the displays will be changing while reading the I/O port. If you are using a software monitor a different value should be obtained each time the I/O port is read. A program may be entered that reads the I/O port and outputs its value to a display device if desired. The I/O port only needs to be written once to select the mode and the digit to be displayed. The digit selected may then be read continuosly.
3. Write a HEX 05 (decimal 5) to the I/O port as in step 1.
4. Normal time display of the seconds units digit is now selected. The digit read from the I/O port should now increment at a 1Hz rate.

5. Additional functions can be tried at this time by selecting the function from Table II-1 and writing the function code to the I/O port. If the board runs steps 1 through 4 successfully, you are ready to begin developing programs for use with the board. Do not install a backup battery option until all your programs that display the time and date are operational.

If the CT102-A is to be accessed by a machine language program remember that a 400 micrsecond delay is needed after outputing a function code to allow time for the board to select the requested digit. Most BASIC interpreters are slow so the delay is not required and may immediately access the digit data after outputting a function code. There are several methods that may be used to convert the BCD data from the CT102-A into a usable form. The BCD data can be directly converted to ASCII by ORing a HEX 30 with the BCD digits as they are read. The ASCII data is usable with display devices and may be used as string data in your programs. If any computation is to be performed with the data from the CT102-A it will be necessary to convert the BCD data into a series of numeric values representing MONTH, DAY, HOUR, MINUTE, and SECOND. This is easily done by multiplying the tens digit BCD value by 10 and adding it to the BCD units digit value. This may be done in BASIC easily and the software examples in this manual use this technique to store the time and date as numeric type parameters. The BASIC PRINT command is used to display the numeric parameters directly. If desired, the numeric values could be converted to string data by BASIC. In this case the digits are read one at a time and the BCD digit value is added to decimal 48. This value is then converted to its equivalent ASCII character using the CHR$ command. String manipulation commands may then be used to format the time and date for display.

If the months tens digit is zero, the clock chip outputs a blank and the 7 segment to BCD converter outputs a HEX OF (decimal 15) when the digit data is read. A value of 15 for this digit must be converted to zero for proper display of the month value. This is done in the software examples given in this manual.

Table II-1 Clock Functions

Decimal Output Value	Hex	Digit Selected	Function Selected
0	00	Hour Tens	Read Time
1	01	Hour Units	" "
2	02	Minute Tens	" "
3	03	Minute Units	" "
4	04	Second Tens	" "
5	05	Second Units	" "
8	08	Month Tens	Read Date
9	09	Month Units	" "
10	0A	Day Tens	" "
11	0B	Day Units	" "
16	10	Hour Tens	Slow Time Set (2Hz Count)
17	11	Hour Units	" " " "
18	12	Minute Tens	" " " "
19	13	Minute Units	" " " "
24	18	Month Tens	Slow Date Set (2Hz Count)
25	19	Month Units	" " " "
26	1A	Day Tens	" " " "
27	1B	Day Units	" " " "
32	20	Hour Tens	Fast Time Set (50Hz Count)
33	21	Hour Units	" " " "
34	22	Minute Tens	" " " "
35	23	Minute Units	" " " "
40	28	Month Tens	Fast Date Set (50Hz Count)
41	29	Month Units	" " " "
42	2A	Day Tens	" " " "
43	2B	Day Units	" " " "
55	37	Reset Time To 00:00:00	
63	3F	Reset Date To 12:00	

TABLE II-2 I/O PORT ADDRESS SELECT

DECIMAL ADDRESS RANGE	SW1 A7 1-16	SW2 A6 2-15	SW3 A5 2-14	SW4 A4 3-13	SW5 A3 5-12	SW6 A2 6-11
0-3	X	X	X	X	X	X
4-7	X	X	X	X	X	
8-11	X	X	X	X		X
12-15	X	X	X	X		
16-19	X	X	X		X	X
20-23	X	X	X		X	
24-27	X	X	X			X
28-31	X	X	X			
32-35	X	X		X	X	X
36-39	X	X		X	X	
40-43	X	X		X		X
44-47	X	X		X		
48-51	X	X			X	X
52-55	X	X			X	
56-59	X	X				X
60-63	X	X				
64-67	X		X	X	X	X
68-71	X		X	X	X	
72-75	X		X	X		X
76-79	X	X	X			
80-83	X		X		X	X
84-87	X		X		X	
88-91	X		X			X
92-95	X		X			
96-99	X			X	X	X
100-103	X			X	X	
104-107	X			X		X
108-111	X			X		
112-115	X				X	X
116-119	X				X	
120-123	X					X
124-127	X					

X = Jumper installed switch on.

TABLE II-2 cont. I/O PORT ADDRESS SELECT

DECIMAL ADDRESS RANGE	SW1 A7 1-16	SW2 A6 2-15	SW3 A5 3-14	SW4 A4 4-13	SW5 A3 5-12	SW6 A2 6-11
128-131		X	X	X	X	X
132-135		X	X	X	X	
136-139		X	X	X		X
140-143		X	X	X		
144-147		X	X		X	X
148-151		X	X		X	
152-155		X	X			X
156-159		X	X			
160-163		X		X	X	X
164-167		X		X	X	
168-171		X		X		X
172-175		X		X		
176-179		X			X	X
180-183		X			X	
184-187		X				X
188-191		X				
192-195			X	X	X	X
196-199			X	X	X	
200-203			X	X		X
204-207			X	X		
208-211			X		X	X
212-215			X		X	
216-219			X			X
220-223			X			
224-227				X	X	X
228-231				X	X	
232-235				X		X
236-239				X		
240-243					X	X
244-247					X	
248-251						X
252-255						

X = Jumper installed or switch on.

III. Software

The Time and Date is accessed by outputing a function code and then reading the BCD digit data that was selected by the funtion code. Table II-1 contains a list of the valid function codes. Output operations will select a function, and input operations will read digit data at any of the four addresses within the selected address range.

Two BASIC program examples are provided in the rear of the manual. The first program shows how to set the time and date as well as read time and date. The second program is intended for displaying the time or date while manually setting the clock (battery backup option installed).

Things to consider when writing your own programs:

1. It takes up to 500 microseconds for digit data to become stable after selecting a digit with a function code. A 1 millisecond delay between the function code output and the digit data input instruction is advisable. Interpreters are usually much slower than this but the delay would be necessary in a machine language routine.
2. The date month tens digit may be a blank due to leading zero blanking performed by U1. The BCD data presented by U5 for a blank will be 15 (hex 0F). This value must be converted back to zero for a proper date value (see line 65 of the first BASIC program).
3. When the minutes count past 59 the hour will be counted by one. It is desirable to initially set the hour to the desired value minus one (see line 165 of the first BASIC program). The same is true of the month and day when setting the date (see line 25 of the first BASIC program). Likewise, the date should be set after the time because the date will be counted when the time counts past 23:59.

4. The day of the month will always count to 31 before advancing the month. The day will have to be advanced at the end of the month having 28, 29, or 30 days.
5. It is possible to read the clock while the digits are counting. An incorrect value will be obtained under these conditions. This will occur most often when reading seconds digits or when setting the clock. To avoid reading bad data, the clock should be read twice, and the compared. If the data is different, it indicates a count has taken place and the data should be read again. When two consecutive reads produce identical data, it gurantees that the data was not read while the clock was counting. The BASIC program examples do not use this technique in order to keep them simple. The mis-read of data occurs so seldom, that it is not a problem when using visual display only. If the clock data is to be recorded permanently on printed or storage media it would be wise to insure its accuracy.

IV. Circuit Description:

The clock board is accessed when comparitor U7 detects the address selected at SW1. If an input I/O operation is being performed the SINP will go high and gate U9 will enable bus drivers U10 and a portion of U11. The top 4 bits of the S100 DI bus (data-in) are provided with zeros and the lower 4 bits are provided with BCD data from U5.

When an I/O output operation is performed, the SOUT signal will go high and a low signal will be presented to pin 2 of U11. This low will enable into the clock input of U8 when the $\overline{\text{PWR}}$ write strobe at pin 77 goes low. When the $\overline{\text{PWR}}$ signal returns high the clock input to U8 will transition high and latch the output data from the S100 DO bus.

The 6 outputs of latch U8 are level shifted to the clock circuit voltages by U3. The lower 3 signals go to U4 which selects a clock digit to be output from the clock chip U1. The digit scan oscillator is stopped on the selected digit by shorting the selected digit output to the junction of R17 and C8 (digit scan oscillator RC network). The scan oscillator will free run until the selected digit is present. This circuitry will select a digit in a maximum of 500 microseconds. It is not necessary to delay reading a digit after it is selected by interpreter type languages such as BASIC, as they are slower than the clock circuitry.

The upper 3 bits of the signal from U8 and U3 are used to control clock functions of U1 such as TIME/DATE select fast set, and slow set. U2 and U6 form a time base circuit which provides the clock circuit U1 with a crystal controlled 50 Hz time reference. C7 is used to trim the crystal frequency for maximum accuracy.

The selected clock digit appears as 7 segment data at U1 and is converted to BCD by U5.

V. Backup Battery Option:

1. Cut etches
   C to D.
   E to F.
   H to J.
2. Connect the negative side of the backup battery to point A and the positive side to point B.
3. NOTE:

   The fast set function is now controlled by switch 7 on SW1 and slow set function is now controlled by switch 8 on SW1. The clock is manually set using these switches. The switches are left open for normal timekeeping operation. An example of a manual set program is included. This program should be run when setting the clock with the backup battery option. Do not attempt to use the software clock set function (etches E to F and H to J uncut) as the clock setting will change when the system is powered down and defeat the purpose of having the battery.

   If it is desired to have battery backup and still set time and date through software, proceed as follows:

   Do steps 1 and 2 above.

   Wire a double pole, single throw switch to pads H, J, E, and F as shown in Figure A-1. You should locate the switch external to the board. Be sure to make the leads long enough when wiring to the pads. Leave the switch in the off position for normal operation to protect the time setting when powering the system on or off. Only turn the switch on when it is desired to set the time or date through software and turn it off when setting is completed.



Figure A-1

Backup Battery Considerations:

Cosiderable blank space exits on the board. Certain battery types can be mounted directly on the circuit board. Do not use battery types that will be damaged by heat from adjacent circuit cards. Also, beaware of batteries that may leak corrosive chemicals into the system.

Lead-acid batteries of the sealed type make excellent backup sources. They have large capacities and can be trickle-charged with simple circuits. Ni-cad batteries do not usually respond well to trickle-charging due to "memory" effects. Ni-cad batteries may lose most of their usable capacity in standby mode. They can fail shortly after they are called upon to backup the clock circuitry. If Ni-cads are used, they should be discharged periodically and re-charged to insure that they retain full capacity.

Do not attempt to trickle-charge carbon or akaline type batteries for long periods of time. Even a small amount of charge current will eventually ruin the battery.

Some calculator manufacturers have removable combination charger and battery packs (Texas Instruments battery kit RK2). These may be mounted on the board and the 110 volt charger can be permanently plugged in seperate from the system so the batteries are charging while the system is powered off.

When using a lead-acid battery, a trickle-charge circuit such as the one shown in figure A-2 can be used to keep the battery charged when the system is powered down.



Figure A-2 Backup battery trickle-charge circuit



Note: If backup supply is used, do the following:

1) Cut etches C to D, E to F, and H to J.
2) Install switches for setting clock:
   Close 7 to 10 for fast set.
   Close 8 to 9 for Slow set.
3) Connect backup supply + to B and − to A

APPENDIX

```
5 GOTO 290
10 INPUT "WHAT IS THE DATE.  MO,DY "; M,D
15 IF M=0 THEN 10
17 IF D=0 THEN 10
20 IF M > 12 THEN 10
25 IF M=1 THEN LET C=12 ELSE LET C=M-1 REMARK SET MONTH-1 INITIALLY
30 IF D > 31 THEN 10
40 PRINT "SETTING DATE"
50 OUT 128,40 REMARK MONTH TENS & FAST SET
60 A=INP(128)*10
65 IF A = 150 THEN LET A = 0
70 OUT 128,41 REMARK MONTH UNITS & FAST SET
80 B=A+INP(128)
85 IF B<>P THEN GOSUB 740
90 IF B<>C THEN 50
100 OUT 128,26 REMARK DAY TENS & SLOW SET
110 A=INP(128)*10
120 OUT 128,27 REMARK DAY UNITS & SLOW SET
130 B=A+INP(128)
135 IF B<>P THEN GOSUB 770
140 IF B<>D THEN 100
141 OUT 128,24 REMARK MONTH TENS & SLOW SET
142 A=INP(128)*10
143 IF A=150 THEN LET A=0
144 OUT 128,25 REMARK MONTH UNITS & SLOW SET
145 B=A+INP(128)
146 IF B<>P THEN GOSUB 740
147 IF B<>M THEN 100
148 OUT 128,0
149 GOTO 290
150 INPUT "WHAT IS THE TIME.  HR,MI  ";H,M
160 IF H>23 THEN 150
165 IF H=0 THEN LET C=23 ELSE LET C=H-1 REMARK SET HOUR-1 INITIALLY
170 IF M>59 THEN 150
180 PRINT "SETTING TIME"
190 OUT 128,32 REMARK HOUR TENS & FAST SET
200 A=INP(128)*10
210 OUT 128,33 REMARK HOUR MINUTES & FAST SET
220 B=A+INP(128)
225 IF B<>P THEN GOSUB 740
230 IF B<>C THEN 190
240 OUT 128,18 REMARK MINUTE TENS & SLOW SET
250 A=INP(128)*10
260 OUT 128,19 REMARK MINUTE UNITS & SLOW SET
270 B=A+INP(128)
275 IF B<>P THEN GOSUB 770
280 IF B<>M THEN 240
281 OUT 128,16 REMARK HOUR TENS & SLOW SET
282 A=INP(128)*10
283 OUT 128,17 REMARK HOUR UNITS & SLOW SET
284 B=A+INP(128)
285 IF B<>P THEN GOSUB 740
286 IF B<>H THEN 240
287 OUT 128,0
288 GOTO 10
290 PRINT "TYPE D TO DISPLAY DATE."
300 PRINT "TYPE T TO DISPLAY TIME."
310 PRINT "TYPE S TO ADVANCE TO NEXT MINUTE & ZERO SECONDS."
320 PRINT "TYPE R TO RESET TIME AND DATE."
330 INPUT "TYPE E TO END THE PROGRAM."; I$
```

```
340 IF I$ = "R" THEN 150
350 IF I$ = "D" THEN 405
360 IF I$ = "T" THEN 515
370 IF I$ = "S" THEN 670
380 IF I$ = "E" THEN 400
390 GOTO 290
400 STOP
405 PRINT
410 PRINT "THE DATE IS ";
420 OUT 128,8 REMARK MONTH TENS
430 A=INP(128)*10
435 IF A = 150 THEN LET A = 0
440 OUT 128,9 REMARK MONTH UNITS
450 M=A+INP(128)
460 OUT 128,10 REMARK DAY TENS
470 A=INP(128)*10
480 OUT 128,11 REMARK DAY UNITS
490 D=A+INP(128)
491 REMARK  THE DATE IS CONVERTED TO 4 YEAR FORMAT.
492 IF D=31 AND (M=2 OR M=4 OR M=6 OR M=9 OR M=11) THEN GOTO 511
493 IF M=2 AND (D=28 OR D=30) THEN GOTO 511
500 PRINT M;":";D
505 PRINT
510 GOTO 290
511 OUT 128,25
512 A=INP(128)
513 B=INP(128)
514 IF A=B THEN GOTO 513 ELSE GOTO 420
515 PRINT
520 PRINT "THE TIME IS ";
530 OUT 128,0 REMARK HOUR TENS
540 A=INP(128)*10
550 OUT 128,1 REMARK HOUR UNITS
560 H=A+INP(128)
570 OUT 128,2 REMARK MINUTE TENS
580 A=INP(128)*10
590 OUT 128,3 REMARK MINUTE UNITS
600 M=A+INP(128)
610 OUT 128,4 REMARK SECOND TENS
620 A=INP(128)*10
630 OUT 128,5 REMARK SECOND UNITS
640 S=A+INP(128)
650 PRINT H;":";M;":";S
655 PRINT
660 GOTO 290
670 OUT 128,19 REMARK MINUTE UNITS & SLOW SET
680 A=INP(128)
690 B=INP(128)
700 IF A=B THEN 690 REMARK CONTINUE UNTIL MINUTES TOGGLE
710 OUT 128,0
720 GOTO 515
740 LET P=B
750 PRINT P
760 RETURN
770 LET P=B
780 PRINT ":";P
790 RETURN
800 END

H>
```

```
BASIC-E COMPILER  VER 2.1
    1:  10 INPUT "TIME OR DATE "; I$
    2*  20 IF I$ = "D" THEN 100
    3*  30 IF I$ = "T" THEN 200
    4*  40 GOTO 10:
    5:  100 OUT 128,8 REMARK MONTH TENS
    6*  110 A=INP(128)*10
    7*  120 IF A=150 THEN LET A=0 REMARK BLANKED LEADING ZERO
    8*  130 OUT 128,9 REMARK MONTH UNITS
    9*  140 M=A+INP(128)
   10*  150 OUT 128,10 REMARK DAY TENS
   11*  160 A=INP(128)*10
   12*  170 OUT 128,11 REMARK DAY UNITS
   13*  180 D=A+INP(128)
   14*  190 PRINT M;":";D
   15*  195 GOTO 100
   16:  200 OUT 128,0 REMARK HOUR TENS
   17*  210 A=INP(128)*10
   18*  220 OUT 128,1 REMARK HOUR UNITS
   19*  230 H=A+INP(128)
   20*  240 OUT 128,2 REMARK MINUTE TENS
   21*  250 A=INP(128)*10
   22*  260 OUT 128,3 REMARK MINUTE UNITS
   23*  270 M=A+INP(128)
   24*  280 OUT 128,4 REMARK SECOND TENS
   25*  290 A=INP(128)*10
   26*  300 OUT 128,5 REMARK SECOND UNITS
   27*  310 S=A+INP(128)
   28*  320 PRINT H;":";M;":";S
   29*  330 GOTO 200
   30*  340 END
    0 ERRORS DETECTED
```

```
10 REM ****** MICROSOFT DISK BASIC, RUNNING UNDER CP/M ******
20 REM *** REV. BY HARRY KAEMMERER  07/14/1979 ***
30 REM *** PGM TO DISPAY TIME AND DATE OF COMPU/TIME MODEL T102A CLOCK ***
40 REM *** SET DISPLAY SCREEN WIDTH ***
50 WIDTH 80
60 REM *** CLEAR SCREEN COMMAND ***
70 PRINT CHR$(12)
80 REM
90 REM *** CHANGE P1= TO DECIMAL ADDRESS OF YOUR STARTING PORT ***
100 REM *** 192=C0Hex ***
110 P1=192
120 H=23:V=0:GOSUB 680
130 PRINT" TIME ";
140 REM *** HOUR TENS ***
150 OUT P1,0
160 H1=INP(P1)
170 REM *** HOUR UNITS ***
180 OUT P1,1
190 H2=INP(P1)
200 REM *** MINUTE TENS ***
210 OUT P1,2
220 M1=INP(P1)
230 REM *** MINUTE UNITS ***
240 OUT P1,3
250 M2=INP(P1)
260 REM *** SECOND TENS ***
270 OUT P1,4
280 S1=INP(P1)
290 REM *** SECOND UNITS ***
300 OUT P1,5
310 S2=INP(P1)
320 REM *** ELIMINATE SPACES BETWEEN NUMBERS BY CONVERTING TO STRINGS ***
330 H1$=CHR$(H1+48)
340 H2$=CHR$(H2+48)
350 M1$=CHR$(M1+48)
360 M2$=CHR$(M2+48)
370 S1$=CHR$(S1+48)
380 S2$=CHR$(S2+48)
390 PRINT H1$;H2$;":";M1$;M2$;":";S1$;S2$;
400 PRINT"  DATE ";
410 REM *** MONTH TENS ***
420 OUT P1,8
430 M1=INP(P1)
440 IF M1=15 THEN M1=0
450 REM *** MONTH UNITS ***
460 OUT P1,9
470 M2=INP(P1)
480 REM *** DAY TENS ***
490 OUT P1,10
500 D1=INP(P1)
510 REM *** DAY UNITS ***
520 OUT P1,11
530 D2=INP(P1)
540 REM *** ELIMINATE SPACES BETWEEN NUMBERS BY CONVERTING TO STRINGS ***
550 M1$=CHR$(M1+48)
560 M2$=CHR$(M2+48)
570 D1$=CHR$(D1+48)
580 D2$=CHR$(D2+48)
```

```
590 REM *** CHANGE NEXT LINE TO CURRENT YEAR ***
600 PRINT M1$;M2$;"/";D1$;D2$;"/1979"
610 PRINT:PRINT
620 END
630 REM ----------------------------
640 REM *** CURSOR CONTROL SUB. ***
650 REM ----------------------------
660 REM
670 REM H=HORZ  V=VERTICAL
680 OUT 1,3
690 A=INP(0) AND 128:IF A<>0 THEN 690
700 OUT 1,H
710 A=INP(0) AND 128:IF A<>0 THEN 710
720 OUT 1,V
730 A=INP(0) AND 128:IF A<>0 THEN 730
740 RETURN
```

```
10 REM ****** MICROSOFT DISK BASIC, RUNNING UNDER CP/M ******
20 REM *** REV. BY HARRY KAEMMERER  07/14/1979 ***
30 REM *** PGM TO RUN COMPU/TIME CLOCK BOARD MODEL T102A ***
40 REM *** SET DISPLAY SCREEN WIDTH ***
50 WIDTH 80
60 REM *** CLEAR SCREEN COMMAND ***
70 PRINT CHR$(12)
80 REM *** CHANGE P1= TO DECIMAL ADDRESS OF YOUR STARTING PORT ***
90 REM *** 192=C0Hex ***
100 P1=192
110 REM *** GOTO START OF MASTER PROGRAM ***
120 GOTO 1270
130 PRINT
140 PRINT"INVALID DATA DO IT AGAIN"
150 PRINT
160 INPUT"WHAT IS THE DATE MO,DY";M,D
170 IF M=0 THEN 130
180 IF D=0 THEN 130
190 IF M>12 THEN 130
200 REM *** SET MONTH -1 INITIALLY ***
210 IF M=1 THEN C=12 ELSE C=M-1
220 IF D>31 THEN 130
230 PRINT"SETTING DATE"
240 REM *** MONTH TENS & FAST SET ***
250 OUT P1,40
260 A=INP(P1)*10
270 IF A=150 THEN A=0
280 REM *** MONTH UNITS & FAST SET ***
290 OUT P1,41
300 B=A+INP(P1)
310 IF B<>P THEN GOSUB 1660
320 IF B<>C THEN 250
330 REM *** DAY TENS & SLOW SET ***
340 OUT P1,26
350 A=INP(P1)*10
360 REM *** DAY UNITS & SLOW SET ***
370 OUT P1,27
380 B=A+INP(P1)
390 IF B<>P THEN GOSUB 1700
400 IF B<>D THEN 340
410 REM *** MONTH TENS & SLOW SET ***
420 OUT P1,24
430 A=INP(P1)*10
440 IF A=150 THEN A=0
450 REM *** MONTH UNITS & SLOW SET ***
460 OUT P1,25
470 B=A+INP(P1)
480 IF B<>P THEN GOSUB 1660
490 IF B<>M THEN 340
500 OUT P1,0
510 GOTO 1270
520 PRINT
530 PRINT"INVALID DATA DO IT AGAIN"
540 PRINT
550 INPUT"WHAT IS THE TIME. HR,MI";H,M
560 IF H>23 THEN 520
570 REM *** SET HOUR -1 INITIALLY ***
580 IF H=0 THEN C=23 ELSE C=H-1
```

```
590 IF M>59 THEN 520
600 PRINT"SETTING TIME"
610 REM *** HOUR TENS & FAST SET ***
620 OUT P1,32
630 A=INP(P1)*10
640 REM *** HOUR MINUTES & FAST SET ***
650 OUT P1,33
660 B=A+INP(P1)
670 IF B<>P THEN GOSUB 1660
680 IF B<>C THEN 620
690 REM *** MINUTES TENS & SLOW SET ***
700 OUT P1,18
710 A=INP(P1)*10
720 REM *** MINUTES UNITS & SLOW SET ***
730 OUT P1,19
740 B=A+INP(P1)
750 IF B<>P THEN GOSUB 1700
760 IF B<>M THEN 700
770 REM *** HOUR TENS & SLOW SET ***
780 OUT P1,16
790 A=INP(P1)*10
800 REM *** HOUR UNITS & SLOW SET ***
810 OUT P1,17
820 B=A+INP(P1)
830 IF B<>P THEN GOSUB 1660
840 IF B<>H THEN 700
850 OUT P1,0
860 GOTO 1270
870 REM **** START OF MASTER PROGRAM ****
880 PRINT
890 PRINT"INVALID DATA DO IT AGAIN"
900 PRINT
910 PRINT"TYPE <T> FOR TIME & DATE."
920 PRINT"TYPE <RT> TO RESET TIME."
930 PRINT"TYPE <RD> TO RESET DATE."
940 PRINT"TYPE <S> TO ADVANCE TO NEXT MINUTE & ZERO SECONDS."
950 INPUT"TYPE <E> TO EXIT PROGRAM";I$
960 IF I$="RT" THEN 540
970 IF I$="RD" THEN 150
980 IF I$="T" THEN 1270
990 IF I$="S" THEN 1590
1000 IF I$="E" THEN 1740
1010 GOTO 880
1020 REM
1030 PRINT
1040 PRINT"  DATE ";
1050 REM *** MONTH TENS ***
1060 OUT P1,8
1070 M1=INP(P1)
1080 IF M1=15 THEN M1=0
1090 REM *** MONTH UNITS ***
1100 OUT P1,9
1110 M2=INP(P1)
1120 REM *** DAY TENS ***
1130 OUT P1,10
1140 D1=INP(P1)
1150 REM *** DAY UNITS ***
1160 OUT P1,11
```

```
1170 D2=INP(P1)
1180 REM *** ELIMINATE SPACES BETWEEN NUMBERS BY CONVERTING TO STRINGS ***
1190 M1$=CHR$(M1+48)
1200 M2$=CHR$(M2+48)
1210 D1$=CHR$(D1+48)
1220 D2$=CHR$(D2+48)
1230 REM *** CHANGE NEXT LINE TO CURRENT YEAR ***
1240 PRINT M1$;M2$;"/";D1$;D2$;"/1979"
1250 PRINT
1260 GOTO 900
1270 PRINT CHR$(12)
1280 PRINT
1290 PRINT" TIME ";
1300 REM *** HOUR TENS ***
1310 OUT P1,0
1320 H1=INP(P1)
1330 REM *** HOUR UNITS ***
1340 OUT P1,1
1350 H2=INP(P1)
1360 REM *** MINUTE TENS ***
1370 OUT P1,2
1380 M1=INP(P1)
1390 REM *** MINUTE UNITS ***
1400 OUT P1,3
1410 M2=INP(P1)
1420 REM *** SECOND TENS ***
1430 OUT P1,4
1440 S1=INP(P1)
1450 REM *** SECOND UNITS ***
1460 OUT P1,5
1470 S2=INP(P1)
1480 REM *** ELIMINATE SPACES BETWEEN NUMBERS BY CONVERTING TO STRINGS ***
1490 H1$=CHR$(H1+48)
1500 H2$=CHR$(H2+48)
1510 M1$=CHR$(M1+48)
1520 M2$=CHR$(M2+48)
1530 S1$=CHR$(S1+48)
1540 S2$=CHR$(S2+48)
1550 PRINT H1$;H2$;":";M1$;M2$;":";S1$;S2$;
1560 GOTO 1040
1570 REM --- SUB. STARTS HERE ---
1580 REM *** MINUTE UNITS & SLOW SET ***
1590 OUT P1,19
1600 A=INP(P1)
1610 B=INP(P1)
1620 REM *** CONTINUE UNTIL MINUTES TOGGLE ***
1630 IF A=B THEN 1610
1640 OUT P1,0
1650 GOTO 1270
1660 P=B
1670 PRINT P
1680 RETURN
1690 REM ---
1700 P=B
1710 PRINT":";P
1720 RETURN
1730 REM *** END OF PROGRAM, CHANGE NEXT LINE TO YOUR END STATEMENT ***
1740 END
```

```
10 REM ****** MICROSOFT DISK BASIC, RUNNING UNDER CP/M ******
20 REM *** REV. BY HARRY KAEMMERER  07/14/1979 ***
30 REM *** PGM TO AUTO DISPLAY & UPDATE TIME OF COMPU/TIME
40 REM      CLOCK MODEL T102A ***
50 REM
60 REM *** SET DISPLAY SCREEN WIDTH ***
70 WIDTH 80
80 REM *** CLEAR SCREEN COMMAND ***
90 PRINT CHR$(12)
100 REM
110 REM *** CHANGE P1= TO DECIMAL ADDRESS OF YOUR STARTING PORT ***
120 REM *** 192=C0Hex ***
130 P1=192
140 X=0
150 H=23:V=0:GOSUB 780
160 IF X=1 THEN PRINT CHR$(8)
170 IF X=0 THEN PRINT" TIME ";
180 REM *** HOUR TENS ***
190 OUT P1,0
200 H1=INP(P1)
210 REM *** HOUR UNITS ***
220 OUT P1,1
230 H2=INP(P1)
240 REM *** MINUTE TENS ***
250 OUT P1,2
260 M1=INP(P1)
270 REM *** MINUTE UNITS ***
280 OUT P1,3
290 M2=INP(P1)
300 REM *** SECOND TENS ***
310 OUT P1,4
320 S1=INP(P1)
330 REM *** SECOND UNITS ***
340 OUT P1,5
350 S2=INP(P1)
360 T1=H1+H2+M1+M2+S1+S2
370 IF T1=T2 THEN 190
380 T2=T1
390 REM *** ELIMINATE SPACES BETWEEN NUMBERS BY CONVERTING TO STRINGS ***
400 H1$=CHR$(H1+48)
410 H2$=CHR$(H2+48)
420 M1$=CHR$(M1+48)
430 M2$=CHR$(M2+48)
440 S1$=CHR$(S1+48)
450 S2$=CHR$(S2+48)
460 IF X=1 THEN H=29:V=0:GOSUB 780
470 PRINT H1$;H2$;":";M1$;M2$;":";S1$;S2$;
480 IF X=1 THEN 160
490 PRINT"  DATE ";
500 REM *** MONTH TENS ***
510 OUT P1,8
520 M1=INP(P1)
530 IF M1=15 THEN M1=0
540 REM *** MONTH UNITS ***
550 OUT P1,9
560 M2=INP(P1)
570 REM *** DAY TENS ***
580 OUT P1,10
```

```
590 D1=INP(P1)
600 REM *** DAY UNITS ***
610 OUT P1,11
620 D2=INP(P1)
630 REM *** ELIMINATE SPACES BETWEEN NUMBERS BY CONVERTING TO STRINGS ***
640 M1$=CHR$(M1+48)
650 M2$=CHR$(M2+48)
660 D1$=CHR$(D1+48)
670 D2$=CHR$(D2+48)
680 REM *** CHANGE NEXT LINE TO CURRENT YEAR ***
690 PRINT M1$;M2$;"/";D1$;D2$;"/1979"
700 PRINT:PRINT
710 X=X+1
720 GOTO 160
730 REM ---------------------------
740 REM *** CURSOR CONTROL SUB. ***
750 REM ---------------------------
760 REM
770 REM H=HORZ  V=VERTICAL
780 OUT 1,3
790 A=INP(0) AND 128:IF A<>0 THEN 790
800 OUT 1,H
810 A=INP(0) AND 128:IF A<>0 THEN 810
820 OUT 1,V
830 A=INP(0) AND 128:IF A<>0 THEN 830
840 RETURN
```

```
10 REM TIMESET2, A PROGRAM TO READ TIME CONTINUOUSLY
20 REM DEVELOPED BY DR. A. L. WINTERS, 12/06/79
30 DIM J$(24)
40 !CHR$(3)\!CHR$(93)
50 P=244\REM PORT ADDRESS
60 OUT P,8
70 C1=INP(P)
80 IF C1=15 THEN C1=0
90 OUT P,9
100 C2=INP(P)
110 OUT P,10
120 D1=INP(P)
130 OUT P,11
140 D2=INP(P)
150 D3=D1*10\D4=D3+D2\C3=C1*10\C4=C3+C2
160 OUT P,0
170 H1=INP(P)
180 OUT P,1
190 H2=INP(P)
200 OUT P,2
210 M1=INP(P)
220 OUT P,3
230 M2=INP(P)
PRESS RETURN TO CONTINUE
240 OUT P,4
250 S1=INP(P)
260 OUT P,5
270 S2=INP(P)
280 H1$=CHR$(H1+48)\H2$=CHR$(H2+48)
290 M1$=CHR$(M1+48)\M2$=CHR$(M2+48)
300 S1$=CHR$(S1+48)\S2$=CHR$(S2+48)
310 D1$=CHR$(D1+48)\D2$=CHR$(D2+48)
320 C1$=CHR$(C1+48)\C2$=CHR$(C2+48)
330 J1$= " DATE "\J2$="/"\J3$="/79   "\J4$="!"
340 J$=J1$+C1$+C2$+J2$+D1$+D2$+J3$+H1$+H2$+J4$+M1$+M2$+J4$+S1$+S2$
350 !J$,\!CHR$(13),
360 IF D4=29 AND C4=2 THEN 370 ELSE 365
365 IF D4=30 AND C4=2 THEN 400 ELSE 430
370 IF Y<>80 OR Y<>84 OR Y<>88 OR Y<>92 OR Y<>96 OR Y<>00 THEN 380 ELSE 160
380 OUT P,27
390 OUT P,27
400 OUT P,27
410 GOTO 160
430 IF D4=31 AND (C4=4 OR C4=6 OR C4=9 OR C4=11) THEN 440 ELSE 160
440 OUT P,27
450 GOTO 160
READY
```